
Appendix 6 Network meta-analysis

WinBUGS code 

Fixed effects binomial likelihood with logit link 

model{                          # *** PROGRAM STARTS 

for(i in 1:ns){                 # LOOP THROUGH STUDIES 

    mu[i] ~ dnorm(0,.0001)      # vague priors for all trial baselines 

    for (k in 1:na[i])  {       # LOOP THROUGH ARMS 

        r[i,k] ~ dbin(p[i,k],n[i,k])    # binomial likelihood 

# model for linear predictor 

        logit(p[i,k]) <- mu[i] + d[t[i,k]] - d[t[i,1]] 

# expected value of the numerators  

        rhat[i,k] <- p[i,k] * n[i,k] 

#Deviance contribution 

        dev[i,k] <- 2 * (r[i,k] * (log(r[i,k])-log(rhat[i,k])) 

             +  (n[i,k]-r[i,k]) * (log(n[i,k]-r[i,k]) - log(n[i,k]-rhat[i,k]))) 

      } 

# summed residual deviance contribution for this trial 

    resdev[i] <- sum(dev[i,1:na[i]]) 

     }    

totresdev <- sum(resdev[])      # Total Residual Deviance 

d[1]<-0    # treatment effect is zero for reference treatment 

# vague priors for treatment effects 

for (k in 2:nt){  d[k] ~ dnorm(0,.0001) } 

for (c in 1:(nt-1)) {  for (k in (c+1):nt) { 

       or[c,k] <- exp(d[k] - d[c]) 

       lor[c,k] <- (d[k]-d[c]) 

      } 

} 

 

# ranking  

for (k in 1:nt) { 

   rk[k] <- rank(d[],k)                           # assumes events are “bad” 

   best[k] <- equals(rk[k],1)                    #calculate probability that 
treat k is best 

} 

}                            

Random effects binomial likelihood with logit link

model{                               # *** PROGRAM STARTS 

for(i in 1:ns){                      # LOOP THROUGH STUDIES 

    w[i,1] <- 0    # adjustment for multi-arm trials is zero for control arm 

    delta[i,1] <- 0             # treatment effect is zero for control arm 

    mu[i] ~ dnorm(0,.0001)           # vague priors for all trial baselines 

    for (k in 1:na[i]) {             # LOOP THROUGH ARMS 

        r[i,k] ~ dbin(p[i,k],n[i,k]) # binomial likelihood 

        logit(p[i,k]) <- mu[i] + delta[i,k]  # model for linear predictor 

        rhat[i,k] <- p[i,k] * n[i,k] # expected value of the numerators  

#Deviance contribution 

        dev[i,k] <- 2 * (r[i,k] * (log(r[i,k])-log(rhat[i,k]))   

            +  (n[i,k]-r[i,k]) * (log(n[i,k]-r[i,k]) - log(n[i,k]-rhat[i,k])))
} 

#  summed residual deviance contribution for this trial 

    resdev[i] <- sum(dev[i,1:na[i]])        

    for (k in 2:na[i]) {             # LOOP THROUGH ARMS 

# trial-specific LOR distributions 

        delta[i,k] ~ dnorm(md[i,k],taud[i,k]) 

# mean of LOR distributions (with multi-arm trial correction) 

        md[i,k] <-  d[t[i,k]] - d[t[i,1]] + sw[i,k] 

# precision of LOR distributions (with multi-arm trial correction) 

        taud[i,k] <- tau *2*(k-1)/k 

# adjustment for multi-arm RCTs 

        w[i,k] <- (delta[i,k] - d[t[i,k]] + d[t[i,1]]) 

# cumulative adjustment for multi-arm trials 

        sw[i,k] <- sum(w[i,1:k-1])/(k-1) 

      } 

  }    

totresdev <- sum(resdev[])           # Total Residual Deviance 

d[1]<-0       # treatment effect is zero for reference treatment 

# vague priors for treatment effects 

for (k in 2:nt){  d[k] ~ dnorm(0,.0001) } 

sd ~ dunif(0,5)     # vague prior for between-trial SD 

tau <- pow(sd,-2)   # between-trial precision = (1/between-trial variance) 

 

# pairwise ORs and LORs for all possible pair-wise comparisons 

for (c in 1:(nt-1)) {  for (k in (c+1):nt) { 

       or[c,k] <- exp(d[k] - d[c]) 

       lor[c,k] <- (d[k]-d[c]) 

      } 

} 

 

# ranking  

for (k in 1:nt) { 

   rk[k] <- rank(d[],k)                           # assumes events are “bad” 

   best[k] <- equals(rk[k],1)                    #calculate probability that 
treat k is best 

} 

 

}       

Fixed effects normal likelihood and identify link

model{                               # *** PROGRAM STARTS 

for(i in 1:ns){                      #   LOOP THROUGH STUDIES 

    mu[i] ~ dnorm(0,.0001)           # vague priors for all trial baselines 

    for (k in 1:na[i]) {             #  LOOP THROUGH ARMS 

        var[i,k] <- pow(se[i,k],2)   # calculate variances 

        prec[i,k] <- 1/var[i,k]      # set precisions 

        y[i,k] ~ dnorm(theta[i,k],prec[i,k]) # binomial likelihood 

# model for linear predictor 

        theta[i,k] <- mu[i] + d[t[i,k]] - d[t[i,1]] 

#Deviance contribution 

        dev[i,k] <- (y[i,k]-theta[i,k])*(y[i,k]-theta[i,k])*prec[i,k] 

      } 

#  summed residual deviance contribution for this trial 

    resdev[i] <- sum(dev[i,1:na[i]])        

  }    

totresdev <- sum(resdev[])            #Total Residual Deviance 

d[1]<-0       # treatment effect is zero for control arm 

# vague priors for treatment effects 

for (k in 2:nt){  d[k] ~ dnorm(0,.0001) } 

for (c in 1:nt-1){ 

 for (k in 2:nt) { 

  IC[c,k] <- d[k] - d[c] 

 } 

 } 

# ranking  

for (k in 1:nt) { 

     rk[k] <- nt + 1- rank(d[],k)                            

   best[k] <- equals(rk[k],1)  

 } 

 }                      

Random effects normal likelihood and identify link

model{                               # *** PROGRAM STARTS 

for(i in 1:ns){                      #   LOOP THROUGH STUDIES 

    w[i,1] <- 0    # adjustment for multi-arm trials is zero for control arm 

    delta[i,1] <- 0             # treatment effect is zero for control arm 

    mu[i] ~ dnorm(0,.0001)           # vague priors for all trial baselines 

    for (k in 1:na[i]) {             #  LOOP THROUGH ARMS 

        var[i,k] <- pow(se[i,k],2)   # calculate variances 

        prec[i,k] <- 1/var[i,k]      # set precisions 

        y[i,k] ~ dnorm(theta[i,k],prec[i,k]) # binomial likelihood 

        theta[i,k] <- mu[i] + delta[i,k]  # model for linear predictor 

#Deviance contribution 

        dev[i,k] <- (y[i,k]-theta[i,k])*(y[i,k]-theta[i,k])*prec[i,k] 

      } 

#  summed residual deviance contribution for this trial 

    resdev[i] <- sum(dev[i,1:na[i]])        

    for (k in 2:na[i]) {             # LOOP THROUGH ARMS 

# trial-specific LOR distributions 

        delta[i,k] ~ dnorm(md[i,k],taud[i,k]) 

# mean of LOR distributions, with multi-arm trial correction 

        md[i,k] <-  d[t[i,k]] - d[t[i,1]] + sw[i,k] 

# precision of LOR distributions (with multi-arm trial correction) 

        taud[i,k] <- tau *2*(k-1)/k 

# adjustment, multi-arm RCTs 

        w[i,k] <- (delta[i,k] - d[t[i,k]] + d[t[i,1]]) 

# cumulative adjustment for multi-arm trials 

        sw[i,k] <- sum(w[i,1:k-1])/(k-1) 

      } 

  }    

totresdev <- sum(resdev[])            #Total Residual Deviance 

d[1]<-0       # treatment effect is zero for control arm 

# vague priors for treatment effects 

for (k in 2:nt){  d[k] ~ dnorm(0,.0001) } 

sd ~ dunif(0,5)     # vague prior for between-trial SD 

tau <- pow(sd,-2)   # between-trial precision = (1/between-trial variance) 

for (c in 1:nt-1){ 

 for (k in 2:nt) { 

  IC[c,k] <- d[k] - d[c] 

 } 

 } 

# ranking  

for (k in 1:nt) { 

     rk[k] <- nt + 1- rank(d[],k)                            

    best[k] <- equals(rk[k],1)  

 } 

}          

